

MPSMD2RES Workshop 6: Two-Way Repeated and Mixed ANOVA

Dr Ian Hocking, Psychology Programme

Contents


1	Overview	1
2	Objectives	1
3	About this document	2
4	Workshop	4
5	Repeated ANOVA	4
6	Mixed ANOVA	20
7	Versions	38
8	References	38

1 Overview

Data files required	 Therapy
Booklet Version	1.0
Format	standard PDF

2 Objectives

This workshop extends your knowledge of Analysis of Variance (ANOVA) to repeated and mixed factorial designs. In a *repeated* design, all factors are repeated. In a *mixed* design, there is at least one *repeated* factor and one *independent* factor.



Research is creating new knowledge.

Neil Armstrong

3 About this document

This document is available in different formats for students who may have accessibility requirements. See [🔗 Versions](#). The system is still being piloted and I'd be interested in your [🔗 feedback](#).

3.1 Tasks and Your Research Journal

Use this booklet in conjunction with your own *Research Journal*, where you will record your workings, thoughts, and other comments related to the exercises. Your Research Journal can take any form, but a Word document might be best; you can copy and paste output from SPSS alongside your notes.

(If you're looking at a non-standard, accessible version of this document, some of the formatting below will be simplified.)

- When I ask you to complete a task, like calculate a mean, it will be formatted like this.

- This is what a Research Journal reminder looks like. I'll use these when asking you to make a note.


3.2 Other Aspects of this Booklet

- This formatting will be used to highlight something important.

Answer

Here I'll provide answers to questions. Note that this version of the document won't be available until after your workshop.

3.3 *Mathematics and Statistics Help*

If you're not confident in your algebra, which is important for dealing with equations, try this  [Introduction to Algebra](#).

3.4 *Answers*

You'll be provided with a second version of this document, containing answers, a few days after your seminar. I'll include SPSS Syntax and possibly SPSS Data files to help you reproduce the correct answers quickly.

When you use menus and dialogue boxes within SPSS to do analyses, SPSS is actually building up a complex command in its native language, syntax, and then running this command. It is feasible for you to access these complex commands yourself. In any dialogue box, the *paste* button will produce the appropriate syntax to do a particular analysis. You can save this syntax as text and run it again at a later date to get the same output. If you want to repeat an analysis quickly, changing bits like variables or type of test, editing syntax is often the best way.

Paste the syntax into an *SPSS syntax window* using *File > New > Syntax*. Highlight the syntax and click the green arrow to make SPSS run the syntax, producing the appropriate output.

It would be a good idea to get used to SPSS Syntax, though I'm not expecting you to use it instead of the graphical, 'point and click' interface.

4 Workshop

Analysis of variance or ‘ANOVA’ is a statistical technique that allows us to examine the differences between mean scores when there are more than two conditions. In this booklet, we will carry out the analysis required for a design with two independent variables (i.e. two factors). In the [first case](#), we will look at an analysis where both factors are *repeated*. In the [second](#), we will look at a *mixed* analysis where one factor is *repeated* and one is *independent*.

In both cases, the ANOVA will tell us three things:

1. Is there a main effect of factor one on the dependent variable (DV)?
2. Is there a main effect of factor two on the DV?
3. Is there an interaction effect of the two factors on the DV?

There’s a lot to get through in this week’s booklet, but it’s important material. Be prepared to continue with the exercises outside of the seminar.

Answer

All data and syntax for this booklet:

- [Repeated ANOVA SPSS data](#)
- [Repeated ANOVA SPSS syntax](#)
- [Mixed ANOVA SPSS data](#)
- [Mixed ANOVA SPSS syntax](#)

5 Repeated ANOVA

We’ll start by revisiting the example I showed you in the lecture.

Students from the University of Statisticsburg were asked to rate the lectures that they received in relation to how much they enjoyed them. Each student surveyed was taking a course in statistics and a course in criminal psychology and were given the survey at the beginning of the course and again ten weeks later at the end. The researcher wanted to know three things:

1. Did people rate one course more positively than the other?
2. Did people rate the course differently from the start to the end of the course?
3. Was the difference in ratings from the start to the end of the course the same for both courses?

Ratings are provided per student in the table below. Responses could range from 0 (hate it) to 10 (love it).

Table 2: Before and after ratings for the statistics and criminal psychology courses.

Statistics Start	Criminal Start	Statistics End	Criminal End
0	5	4	5
2	7	5	7
1	8	3	6
2	4	2	5
1	5	5	5
0	6	7	6
1	7	5	6
1	5	4	5
1	8	1	7
2	4	5	4

5.1 Step One: Load the data into SPSS

1. Set up a new SPSS data file and enter the data in the table above.
 - Call the variables *stats_start*, *criminal_start*, *stats_end* and *criminal_end*
 - Give them appropriate labels
 - Make sure SPSS knows the data type associated with each (which should be *scale*)

5.2 Step Two: Run descriptive statistics

Before we carry out the ANOVA it is a good idea to get an idea of what is going on with the data.

2. Calculate the mean scores for the four conditions by going to *Analyze > Descriptive statistics > Descriptives*. Put all four conditions into the right hand box and click *OK*.

1. Describe what the means tell you.

Answer

These means represent the average ratings for Statistics and for Criminal Psychology at the beginning and end of the courses. They show that Statistics had much lower ratings than Criminal Psychology at the beginning of the term but the difference was much smaller by the end of the term. Whereas Statistics increased, Criminal Psychology decreased slightly.

SPSS Syntax:


```
DESCRIPTIVES VARIABLES=stats_start criminal_start
stats_end criminal_end
/STATISTICS=MEAN STDDEV MIN MAX.
```

5.3 Step Three: Set up the ANOVA

By looking at the descriptive statistics you have some idea of what students thought about the courses and whether their opinions became more or less positive by the end of the course. Now let's see whether those differences are significant or not.

Read through the instructions below before doing them yourself.

1. Remember that all the SPSS statistical analyses are under the *Analyze* menu, so go there, select *General Linear Model*, then *Repeated Measures*. A window will appear where you can define the repeated measures factors (independent variables). As we have two repeated measures factors, we will define two.
2. First type 'time' into the box next to Within-Subject Factor Name. This variable has two levels (the start and the end of term) so type 2 into the box next to *Number of Levels*. Then click on *Add* and it will show up in the main box. Next, define the second repeated measures variable: 'course'. This also has two levels (statistics and criminal psychology). Once this has been defined and added you should have something that looks like Figure 1.
3. Now click on *Define* and a new box will appear, which should look like Figure 2. This is where we tell the SPSS which columns of data to look at.



The image shows the 'Repeated Measures Define Factor(s)' dialog box in SPSS. The window title is 'Repeated Measures Define Factor(s)'. It contains two main sections: 'Within-Subject Factor Name:' and 'Measure Name:'. Each section has a text input field, a 'Number of Levels:' spinner, and a list box with 'Add', 'Change', and 'Remove' buttons. The 'Within-Subject Factor Name' list box contains 'time(2)' and 'course(2)'. The 'Measure Name' list box is empty. At the bottom are 'Help', 'Reset', 'Cancel', and 'Define' buttons.

Repeated Measures Define Factor(s)

Within-Subject Factor Name:

Number of Levels:

Add Change Remove

time(2)
course(2)

Measure Name:

Add Change Remove

Help Reset Cancel Define

Figure 1: SPSS Repeated Measures Define Factors Dialogue




Figure 2: SPSS Repeated Measures Main ANOVA Dialogue

- Pay close attention to the numbers in Figure 2's *Within-Subjects Variables (time, course)* box. Each line in this box refers to the *column* in the *Data View* where SPSS can find the data for a particular factor combination. (You can think of these combinations as the cells in the overall table when you put ANOVA into tabular form, with marginal means.) For instance, $__?(1,1)$ is SPSS's way of requesting the data that matches level 1 of the first factor and level 1 of the second factor. Likewise, $__?(1,2)$ requests level 1 of the first factor and level 2 of the second. Bearing in mind that factor one is Time (start vs. end) and factor two is Course (statistics vs. criminal psychology):

- stats_start is (1,1)
- criminal_start is (1,2)
- stats_end (2,1)
- criminal_end (2,2)

4. Drag the variables across to the *Within-Subjects Variables (time, course)* box. It should look like Figure 3.




Figure 3: SPSS Repeated Measures Main ANOVA Dialogue, completed

5. Make sure you also ask SPSS to create a graph of the results by clicking on *Plots*. Put 'Time' along the x-axis (category axis) and enter 'Course' as separate lines. See Figure 4. Now click *Add* then *Continue*.
6. Lastly, click on *Options* and move everything in the left-hand box into the right-hand box; also tick *Descriptives* and *Homogeneity Tests* (see Figure 5). Click on *Continue* and *OK*.

Now we're ready to run the ANOVA.

2. Go ahead and run the ANOVA as described above.




Figure 4: SPSS Repeated Measures Plots

Answer

SPSS syntax for the ANOVA:

```
* Repeated measures ANOVA (Time by Course).
GLM stats_start criminal_start stats_end criminal_end
  /WSFACTOR=time 2 Polynomial course 2 Polynomial
  /METHOD=SSTYPE(3)
  /PLOT=PROFILE(time*course)
  /EMMEANS=TABLES(OVERALL)
  /EMMEANS=TABLES(time)
  /EMMEANS=TABLES(course)
  /EMMEANS=TABLES(time*course)
  /PRINT=DESCRIPTIVE
  /CRITERIA=ALPHA(.05)
  /WSDSIGN=time course time*course.
```




Figure 5: SPSS Repeated Measures Options

5.4 Step Four: Interpret the Output

We are now ready to answer the important question. Are those differences significant or not?

The table that we should look at first is *Mauchly's Test of Sphericity*. You should have one that looks like Figure 6.

Mauchly's Test of Sphericity^a

Measure: MEASURE_1

Within Subjects Effect	Mauchly's W	Approx. Chi-Square	df	Sig.	Epsilon ^b		
					Greenhouse-Geisser	Huynh-Feldt	Lower-bound
time	1.000	.000	0	.	1.000	1.000	1.000
course	1.000	.000	0	.	1.000	1.000	1.000
time * course	1.000	.000	0	.	1.000	1.000	1.000

Tests the null hypothesis that the error covariance matrix of the orthonormalized transformed dependent variables is proportional to an identity matrix.

a. Design: Intercept
Within Subjects Design: time + course + time * course

b. May be used to adjust the degrees of freedom for the averaged tests of significance. Corrected tests are displayed in the Tests of Within-Subjects Effects table.

Figure 6: Output—Mauchly's Test

2. What do you make of the Mauchly's Test results?

Answer

- Mauchly's test cannot be completed without at least one factor with more than two levels. Because both our factors have two levels each, this table will remain blank.
- When a factor has two levels, we don't actually need to worry about sphericity (the property being assessed by Mauchly's test) because the data happen to have perfect sphericity in this case. To find out more, take a look at Field (2009).

Next, let's look at the *Tests of Within-Subjects Effects*, Figure 7.

- Answer the following questions in plain English and back them up with the appropriate statistics written in APA format. Remember, you will also need to look at the descriptive statistics to interpret

Tests of Within-Subjects Effects

Measure: MEASURE_1

Source		Type III Sum of Squares	df	Mean Square	F	Sig.
time	Sphericity Assumed	18.225	1	18.225	14.232	.004
	Greenhouse-Geisser	18.225	1.000	18.225	14.232	.004
	Huynh-Feldt	18.225	1.000	18.225	14.232	.004
	Lower-bound	18.225	1.000	18.225	14.232	.004
Error(time)	Sphericity Assumed	11.525	9	1.281		
	Greenhouse-Geisser	11.525	9.000	1.281		
	Huynh-Feldt	11.525	9.000	1.281		
	Lower-bound	11.525	9.000	1.281		
course	Sphericity Assumed	99.225	1	99.225	36.413	.000
	Greenhouse-Geisser	99.225	1.000	99.225	36.413	.000
	Huynh-Feldt	99.225	1.000	99.225	36.413	.000
	Lower-bound	99.225	1.000	99.225	36.413	.000
Error(course)	Sphericity Assumed	24.525	9	2.725		
	Greenhouse-Geisser	24.525	9.000	2.725		
	Huynh-Feldt	24.525	9.000	2.725		
	Lower-bound	24.525	9.000	2.725		
time * course	Sphericity Assumed	27.225	1	27.225	23.280	.001
	Greenhouse-Geisser	27.225	1.000	27.225	23.280	.001
	Huynh-Feldt	27.225	1.000	27.225	23.280	.001
	Lower-bound	27.225	1.000	27.225	23.280	.001
Error (time*course)	Sphericity Assumed	10.525	9	1.169		
	Greenhouse-Geisser	10.525	9.000	1.169		
	Huynh-Feldt	10.525	9.000	1.169		
	Lower-bound	10.525	9.000	1.169		

Figure 7: Output—Tests of Within-Subjects Effects

the findings fully.

4. Did people rate one course more positively than the other? Is there a need for further analysis?
5. Did people rate the courses differently from the start to the end of the course? And is there a need for further analysis?
6. Was the difference in ratings from the start to the end of the courses the same for both courses? Is there a need for further analysis?

Answer

4. Yes, the main effect of Course was significant: $F(1,9) = 36.41$, $MSE = 2.73$, $p < .001$. The means tell us that Criminal Psychology was rated more positively than Statistics. Because this factor has only two groups, there is no scope for further analysis.
5. Yes, there was a significant main effect of Time: $F(1,9) = 14.23$, $MSE = 1.28$, $p = .004$. The descriptive statistics tell us that ratings, averaged across both courses, were higher at the end of the courses than at the beginning. Since there are only two time points, there is no need for further analysis.
6. The difference in ratings was not the same for both courses, since the interaction between Course and Time was significant: $F(1,9) = 23.28$, $MSE = 1.17$, $p = .001$. We need to explore this with further analysis.

3. As you did last time, make the interaction graph look like the one in Figure 8 by using the SPSS graph editor.




Figure 8: Output—Profile Plots

7. Generally, and with regards to a potential interaction, what does the graph show?

Answer

The lines are not parallel, which might indicate the interaction is significant. The pattern of cell means would indicate that the ratings for Statistics changed more than the ratings for Criminal between the start and end.

Since the interaction was significant, we'll need to carry out Simple Main Effects analyses. To do this, you'll need to carry out one-way repeated measures ANOVAs for each factor within the levels of the other factor. Carry out these analyses using the instructions in the lecture slides and answer the questions below. Remember that this analysis is very similar to one you've just done with two factors, but now you'll have one factor (for instance, Course) constrained to one level of the other factor (for instance, Time 1).

8. Were the Simple Main Effects of Course within each level of Time significant? Report the statistics to back up your response and explain what these results mean in plain English.

Answer

- The SME of Course was significant at the Start: $F(1,9) = 70.05$, $MSE = 1.64$, $p < .001$. The means show that Criminal Psychology was rated more positively than Statistics.
- The SME of Course was not significant at the End: $F(1,9) = 5.00$, $MSE = 2.25$, $p = .052$, which means that by the end of term, students rated Statistics and Criminal Psychology equally. (Alternatively, this can be treated as a marginally

significant difference, showing that Criminal was still liked better than Statistics, although the gap is much narrower than at the start of term.)

SPSS syntax for the above ANOVAs:

* SME of Course for Time1 (beginning).

```
GLM stats_start criminal_start
  /WSFACTOR=Course_time1 2 Polynomial
  /METHOD=SSTYPE(3)
  /PLOT=PROFILE(Course_time1)
  /EMMEANS=TABLES(OVERALL)
  /EMMEANS=TABLES(Course_time1)
  /PRINT=DESCRIPTIVE
  /CRITERIA=ALPHA(.05)
  /WSDESIGN=Course_time1.
```

* SME of Course for Time2 (end).

```
GLM stats_end criminal_end
  /WSFACTOR=Course_time2 2 Polynomial
  /METHOD=SSTYPE(3)
  /PLOT=PROFILE(Course_time2)
  /EMMEANS=TABLES(OVERALL)
  /EMMEANS=TABLES(Course_time2)
  /PRINT=DESCRIPTIVE
  /CRITERIA=ALPHA(.05)
  /WSDESIGN=Course_time2.
```

9. Were the Simple Main Effects of Time within each level of Course significant? Report the statistics to back up your response and

explain what these results mean in plain English.

Answer

- The SME of Time was significant for Statistics: $F(1,9) = 21.32$, $MSE = 2.11$, $p = .001$. The means show that ratings increased for Statistics from the start to the end of the course.
- However, the SME of Time was not significant within Criminal Psychology: $F(1,9) = 1.33$, $MSE = 0.34$, $p = .279$. This means there was no change in ratings between the start and the end of the course for Criminal Psychology.

SPSS syntax for the above ANOVAs:

* SME of Time for Course1 (statistics).

```
GLM stats_start stats_end
  /WSFACTOR=Time_Course1 2 Polynomial
  /METHOD=SSTYPE(3)
  /PLOT=PROFILE(Time_Course1)
  /EMMEANS=TABLES(Time_Course1)
  /EMMEANS=TABLES(OVERALL)
  /PRINT=DESCRIPTIVE
  /CRITERIA=ALPHA(.05)
  /WSDESIGN=Time_Course1.
```

* SME of Time for Course2 (criminal psychology).

```
GLM criminal_start criminal_end
  /WSFACTOR=Time_Course2 2 Polynomial
  /METHOD=SSTYPE(3)
  /PLOT=PROFILE(Time_Course2)
  /EMMEANS=TABLES(Time_Course2)
```

```

/EMMEANS=TABLES(OVERALL)
/PRINT=DESCRIPTIVE
/CRITERIA=ALPHA(.05)
/WSDESIGN=Time_Course2.

```

6 Mixed ANOVA

So far you have looked at ANOVAs where all independent variables have been of one design or another (either repeated measures/within-groups or independent/between-groups). Now we will work through the analysis required when you have two factors (two-way), one of which has participants contributing to all conditions (repeated measures) while the other has different participants in each group (independent measures). We call this design *mixed*.

As always, the ANOVA will tell us three things:

1. Is there a main effect of IV1 on the DV?
2. Is there a main effect of IV2 on the DV?
3. Is there an interaction effect of the two IVs on the DV?

In the following analysis, we'll build upon an example you've looked at before.

A researcher put 30 depressed patients into one of three therapy groups (CBT, Sunshine therapy, no therapy). She found that having some therapy had a positive effect on sociability levels but it didn't matter whether that therapy was CBT or sunshine. Three months later the researcher was interested in finding out whether the effects of therapy had lasted so she got the group of patients together again and recorded, just like the first time, the amount of time each patient engaged in social activities.

6.1 Step One: Load the data into SPSS

4. Download the  data file.

6.2 Step Two: Run descriptive statistics

As usual, let's get an overview of what's going on with the data.

5. Calculate mean scores for the three groups. Do this by going to *Analyze > Compare Means > Means*. Put 'activity 1' and 'activity 2' in the *Dependent* box and 'group' into the *Independent box*. Now click *OK*.

Answer

SPSS syntax:

```
MEANS TABLES=activity1 activity2 BY group
/CELLS=MEAN COUNT STDDEV.
```

You should get Figure 9.

10. On the basis of the descriptives, do you think the effect of therapy varies over time?

Answer

Although the overall amount of sociability between time 1 and time 2 hasn't changed, there are differences depending on the group. Activity in the group receiving sunshine therapy has decreased from 2.50 to 1.10 hours, in the CBT group sociability increased from 2.80 to 4 hours, and in the no therapy group it remained more or less stable.

Report

Therapy Group		Sociability level at time 1	Sociability level at time 2
No therapy	Mean	1.0000	1.200
	N	10	10
	Std. Deviation	1.05409	1.1353
CBT	Mean	2.8000	4.000
	N	10	10
	Std. Deviation	.91894	1.0541
Sunshine therapy	Mean	2.5000	1.100
	N	10	10
	Std. Deviation	.84984	.8756
Total	Mean	2.1000	2.100
	N	30	30
	Std. Deviation	1.21343	1.6887

Figure 9: Therapy and Sociability Descriptives

6.3 Step Three: Set up the ANOVA

We'll now see whether the differences in the sample can be said to be differences in the population. To do this, we'll use the same *Repeated Measures* SPSS menu option as before, because this dialogue box allows us to specify not only within-groups factors but between-groups factors as well. (In other words, don't look for an SPSS procedure specifically labelled 'mixed'.)

Read through the instructions below before doing them yourself.

1. Go to *Analyze > General Linear Model > Repeated Measures*. A window will appear where you can define the repeated measures factor (independent variable) you have. As we have only one repeated measures factor (Time: immediately after therapy and three months later), we will just define this one. If you can't remember how to define your repeated measures factor, have a look at [the repeated measures example](#).
2. Your next window, where you define factors, should look like Figure 10.




Figure 10: SPSS Repeated Measures Define Factors Dialogue

- Put the two repeated measures conditions into the *Within-Subjects Variable* box and the independent measures factor into the *Between-Subjects Factor* box. This should look like Figure 11.




Figure 11: SPSS Repeated Measures Factor Selection

- As you've done several times by now, click on *Options* ask for means for every effect. This will give us more detailed descriptive statistics. Click on *Continue*.
- Ask SPSS for two graphs: one with 'Therapy Group' on the horizontal axis and 'Time' as separate lines; and one with 'Time' on the horizontal axis and 'Therapy Group' as separate lines. Click *Continue* and *OK*. You've now set up a basic two-way mixed ANOVA.

5. Carry out the ANOVA according to the method above.

Answer

SPSS syntax for the ANOVA:

```
* Mixed (Therapy Group by Time) ANOVA.
GLM activity1 activity2 BY group
  /WSFACTOR=time 2 Polynomial
  /METHOD=SSTYPE(3)
  /PLOT=PROFILE(group*time time*group)
  /EMMEANS=TABLES(OVERALL)
  /EMMEANS=TABLES(group)
  /EMMEANS=TABLES(time)
  /EMMEANS=TABLES(group*time)
  /PRINT=DESCRIPTIVE
  /CRITERIA=ALPHA(.05)
  /WSDESIGN=time
  /DESIGN=group.
```

6.4 Step Four: Interpret the Output

Before we look at the basic statistics, let's look at the graphs produced by SPSS, Figure 12 and Figure 13. I've suggested you produce two of them because each represents a way of looking at the data. It could well be that you find one arrangement more intuitive than another; be aware, however, that they depict the same information.

10. What can we infer from the graph in terms of main and interaction effects? Explain how you arrived at your answer.




Figure 12: Output—Plot (Therapy by Time)




Figure 13: Output—Plot (Time by Therapy)

Answer

- Since the lines are not parallel, we might expect an interaction.
- The effect of Time would not appear to be significant (averaging data points across each line, there wouldn't be much of a difference between the two marginal means).
- There seems to be an effect of Therapy, as the marginal means for each group (average of Time 1 & Time 2 means for each Therapy group), are different.

Let's see if your inferences are backed up by the ANOVA. Again, there are a number of tables we don't need to look at to understand our findings. The first table that we should look at is *Mauchly's Test of Sphericity*. You should have one that looks like Figure 14.

Mauchly's Test of Sphericity^a

Measure: MEASURE_1

Within Subjects Effect	Mauchly's W	Approx. Chi-Square	df	Sig.	Epsilon ^b		
					Greenhouse-Geisser	Huynh-Feldt	Lower-bound
time	1.000	.000	0	.	1.000	1.000	1.000

Tests the null hypothesis that the error covariance matrix of the orthonormalized transformed dependent variables is proportional to an identity matrix.

a. Design: Intercept + group
Within Subjects Design: time

b. May be used to adjust the degrees of freedom for the averaged tests of significance. Corrected tests are displayed in the Tests of Within-Subjects Effects table.

Figure 14: Output—Mauchly's Test of Sphericity

11. What assumption does the above table relate to? Why are there no significance values?

Answer

The table relates to the Sphericity assumption. There are no significance values because it only applies when the within-subjects factors have more than two levels, which is not the case here.

Next, let's look at the *Tests of Within-Subjects Effects*, Figure 15. This will tell us about the main effect of our repeated measures factor (Time) and the interaction between Time and Therapy Group.

Tests of Within-Subjects Effects

Measure: MEASURE_1

Source		Type III Sum of Squares	df	Mean Square	F	Sig.
time	Sphericity Assumed	.000	1	.000	.000	1.000
	Greenhouse-Geisser	.000	1.000	.000	.000	1.000
	Huynh-Feldt	.000	1.000	.000	.000	1.000
	Lower-bound	.000	1.000	.000	.000	1.000
time * group	Sphericity Assumed	17.200	2	8.600	21.500	.000
	Greenhouse-Geisser	17.200	2.000	8.600	21.500	.000
	Huynh-Feldt	17.200	2.000	8.600	21.500	.000
	Lower-bound	17.200	2.000	8.600	21.500	.000
Error(time)	Sphericity Assumed	10.800	27	.400		
	Greenhouse-Geisser	10.800	27.000	.400		
	Huynh-Feldt	10.800	27.000	.400		
	Lower-bound	10.800	27.000	.400		

Figure 15: Output—Tests of Within-Subjects Effects

- Answer the following questions in plain English and back them up with the appropriate statistics written in APA format. Remember, you will also need to look at the descriptive statistics to interpret the findings fully.
12. Was there an effect of time on sociability? Is there a need for further analysis?
 13. Was there an interaction between time and therapy on sociability? Is there a need for further analysis?

Answer

- No, there was no effect of time on sociability – this means that participants did not engage in more social activity in time 1 than in time 2 or vice versa: $F(1,27) < 0.001$, $MSE = 0.40$, $p > .99$. Since the effect is not significant, there is no need for further analysis. And, even if it were significant,

there is no need to track down where the difference might lie because the factor has only two levels.

- Yes, the interaction between time and therapy was significant, $F(2,27) = 21.50$, $MSE = 0.40$, $p < .001$. This means that there was a different amount of change in levels of activity between time 1 and time 2, depending on the therapy group. Because the interaction is significant, we need Simple Effects analysis to help us determine where the differences across time in Figure 13 actually differ.

The last thing to find out is whether there was an effect of therapy on sociability generally, regardless of how long after therapy the measure was taken. To find this out come down to the table titled *Tests of Between-Subjects Effects*, shown in Figure 16.

Tests of Between-Subjects Effects

Measure: MEASURE_1
Transformed Variable: Average

Source	Type III Sum of Squares	df	Mean Square	F	Sig.
Intercept	264.600	1	264.600	170.914	.000
group	55.600	2	27.800	17.957	.000
Error	41.800	27	1.548		

Figure 16: Output—Tests of Between-Subjects Effects

14. Was there a main effect of therapy on sociability? Is there a need for further analysis?

Answer

Yes, there was a difference in sociability levels depending on the group: $F(2) = 17.96$, $MSE = 1.55$, $p < .001$. We need to follow this up with post-hoc comparisons because we have three groups, so we don't know which mean differs from which.

In this example, we have a main effect of Therapy Group and an interaction between Therapy and Time. We need, therefore, to carry out *post hoc analyses*.

6.5 Investigating the Main Effect of Therapy Group

Therapy is a three-level factor (CBT, sunshine and no therapy). Our main effect tells us that there is a significant difference between at least one pair of levels, or groups (i.e. CBT v. sunshine, CBT v. no therapy, or sunshine v. therapy). Because we're looking for paired differences, we need to do *pairwise comparisons*.

6. Run the pairwise comparisons on Therapy Group to investigate the main effect.
 - Go back into *General Linear Model > Repeated Measures* and click *Define*. (The ANOVA should still be set up if you haven't exited SPSS.)
 - Go to *Options* and ask for the marginal means of 'Group' only. (You can remove the other effects and the 'overall' mean by highlighting them and clicking on the arrow between the two boxes).
 - Click on *compare main effects* and selected *Bonferroni* from the drop-down list immediately below. (This will offer pair-wise comparisons for the means of the levels of 'Group', controlling for family-wise error rate.) Click on *Continue*. Your dialogue box should look like Figure 17. Click *OK*.

Answer

SPSS syntax for the above analysis:

```
GLM activity1 activity2 BY group
```




Figure 17: Repeated Measures ANOVA Dialogue, Pairwise Comparisons for Therapy Group

```

/WSFACTOR=time 2 Polynomial
/METHOD=SSTYPE(3)
/PLOT=PROFILE(group*time time*group)
/EMMEANS=TABLES(group) COMPARE ADJ(BONFERRONI)
/PRINT=DESCRIPTIVE HOMOGENEITY
/CRITERIA=ALPHA(.05)
/WSDESIGN=time
/DESIGN=group.

```

Your *Pairwise Comparisons* table should look like Figure 18.

Pairwise Comparisons

Measure: MEASURE_1

(I) Therapy Group	(J) Therapy Group	Mean Difference (I-J)	Std. Error	Sig. ^b	95% Confidence Interval for Difference ^b	
					Lower Bound	Upper Bound
No therapy	CBT	-2.300 [*]	.393	.000	-3.304	-1.296
	Sunshine therapy	-.700	.393	.259	-1.704	.304
CBT	No therapy	2.300 [*]	.393	.000	1.296	3.304
	Sunshine therapy	1.600 [*]	.393	.001	.596	2.604
Sunshine therapy	No therapy	.700	.393	.259	-.304	1.704
	CBT	-1.600 [*]	.393	.001	-2.604	-.596

Based on estimated marginal means

^a. The mean difference is significant at the

^b. Adjustment for multiple comparisons: Bonferroni.

Figure 18: Output—Pairwise Comparisons for Levels within Therapy Group

15. What were the results of the post hoc tests on the main effect of Therapy? Report the statistics to back up your answer.

Answer

CBT differed from both no therapy ($p < .01$) and sunshine therapy conditions ($p < .01$). Looking at the means, we can see that CBT resulted in higher sociability than the other two. CBT is, therefore, driving the main effect that we see.

6.6 Investigating the Interaction of Therapy Group and Time

In our mixed analysis, we have an independent factor (Therapy Group) and a repeated factor (Time). For various reasons that we don't need to go into—[§ unless you're interested](#); cf. Howell (1997)), it is quite straightforward to examine the interaction in terms of the repeated factor, but more difficult to examine it in terms of the independent factor (and we do not expect you to do this in your assignment). For this reason, we'll focus on looking at the interaction from the perspective of the repeated factor.

This means we'll be asking 'What is the effect of Time within each Therapy Group?' In other words, we're looking at how the repeated factor explains variability within each level of the independent factor. In other words, again, we're going to run three separate one-way repeated measures ANOVAs, presented below as the analyses followed by the question each analysis answers:

1. Time (for No Therapy): Did the No Therapy Group change over time?
2. Time (for CBT): Did the CBT group change over time?
3. Time (for Sunshine): Did the Sunshine group change over time?

At this point, it might be useful to touch upon the analysis we *won't* be doing, i.e. 'What is the effect of Therapy Group within Time?' This analysis, from the perspective of the independent/between-groups factor, would have allowed us examine:

1. Therapy Group (for Time 1): Did Therapy Groups differ at Time 1?
2. Therapy Group (for Time 2): Did Therapy Groups differ at Time 2?

In this case, such an analysis would be theoretically less interesting for us, since we're really interested in the effects of therapy *across* time,

which best answered by the three one-way repeated ANOVAs above (Time (for No Therapy) etc.).

- Again, if you find yourself needing to do Simple Main Effects from the perspective of the between-groups factor, [see this webpage](#), or Howell (1997).

Let's now do the Simple Main Effects analysis of Therapy Group from the perspective of Time. At the moment, our SPSS data file has all cases (participants) selected. We want to focus on each level of Therapy Group and perform an analysis on each. So, because Therapy Group is a between-groups factor, marking group membership on a case-by-case basis, we can simply use *Select Cases* to identify, or filter, the participants we're interested in.

- Read through the instructions for selecting cases and running each ANOVA before performing them.

1. Start by going to *Data > Select Cases*.
2. We're now going to describe our filter to SPSS. Select *If condition is satisfied*.
3. Define the Therapy Group to be selected. Start with *If Group = 1*. Your dialogue box should look like Figure 19.
4. Verify that the selection is working as you expect. Check that the *Data View* shows the appropriate cases to be *selected* (row number is untouched) and *unselected* (row number is struck out). When the filter is 'Therapy Group = 1', you should see something like Figure 20.
5. Now that the scope of the data has been restricted to those participants in the No Therapy group, we can run a one-way repeated




Figure 19: Define Cases where Therapy Group = 1, No Therapy

The image shows the 'Data View' of an SPSS dataset. The first column, 'group', has a value of 1.00 for all rows. The second column, 'activity1', has values ranging from 1.00 to 2.00. The third column, 'activity2', has values ranging from 0.00 to 3.00. The fourth column, 'filter_\$', has values of 1 for rows 1-10 and 0 for rows 11-16. The last four columns are labeled 'var'.

	group	activity1	activity2	filter_\$	var	var	var	var
1	1.00	2.00	2.0	1				
2	1.00	3.00	3.0	1				
3	1.00	1.00	3.0	1				
4	1.00	.00	1.0	1				
5	1.00	.00	.0	1				
6	1.00	.00	.0	1				
7	1.00	1.00	.0	1				
8	1.00	1.00	1.0	1				
9	1.00	2.00	1.0	1				
10	1.00	.00	1.0	1				
11	2.00	1.00	2.0	0				
12	2.00	3.00	3.0	0				
13	2.00	4.00	5.0	0				
14	2.00	3.00	4.0	0				
15	2.00	2.00	3.0	0				
16	2.00	3.00	4.0	0				

Figure 20: Data View with only Therapy Group = 1, No Therapy

measures ANOVA (the factor being Time) and our F ratio will tell us whether there is an effect of time within the No Therapy group.

7. Using the *Select Cases* procedure above, and referring to an earlier workshop booklet on one-way repeated measure ANOVA, go ahead and run one-way repeated ANOVAs of Time for each of the three levels of Therapy Group.
- Tip: If you paste the syntax from the *Select Cases* command, as well as the ANOVA, you may find changing the group variable and running the syntax much faster than doing everything manually via the menu system.

Answer

SPSS syntax for the above analysis:

```
* Examining Simple Main Effects of Time
* within levels of Therapy Group.

* Start by defining our filter to select
* case where Group = 1 (no therapy).
USE ALL.
COMPUTE filter_$(group=1).
VARIABLE LABELS filter_$ 'group=1 (FILTER)'.
VALUE LABELS filter_$ 0 'Not Selected' 1 'Selected'.
FORMATS filter_$ (f1.0).
FILTER BY filter_$.
EXECUTE.

* Repeated measures ANOVA, one way,
* examining effect of Time within
* Therapy Group = 1.
GLM activity1 activity2
```

```

/WSFACTOR=time 2 Polynomial
/METHOD=SSTYPE(3)
/PRINT=DESCRIPTIVE
/CRITERIA=ALPHA(.05)
/WSDESIGN=time.

```

```

* Select cases for Therapy Group = 2.
USE ALL.
COMPUTE filter_$(group=2).
VARIABLE LABELS filter_$ 'group=2 (FILTER)'.
VALUE LABELS filter_$ 0 'Not Selected' 1 'Selected'.
FORMATS filter_$ (f1.0).
FILTER BY filter_$.
EXECUTE.

```

```

* Repeated measures ANOVA, one way,
* examining effect of Time within
* Therapy Group = 2.
GLM activity1 activity2
  /WSFACTOR=time 2 Polynomial
  /METHOD=SSTYPE(3)
  /PRINT=DESCRIPTIVE
  /CRITERIA=ALPHA(.05)
  /WSDESIGN=time.

```

```

* Select cases for Therapy Group = 3.
USE ALL.
COMPUTE filter_$(group=3).

```

```

VARIABLE LABELS filter_$ 'group=3 (FILTER)'.
VALUE LABELS filter_$ 0 'Not Selected' 1 'Selected'.
FORMATS filter_$ (f1.0).
FILTER BY filter_$.
EXECUTE.

* Repeated measures ANOVA, one way,
* examining effect of Time within
* Therapy Group = 3.
GLM activity1 activity2
  /WSFACTOR=time 2 Polynomial
  /METHOD=SSTYPE(3)
  /PRINT=DESCRIPTIVE
  /CRITERIA=ALPHA(.05)
  /WSDESIGN=time.

```

16. Report what the above analysis tells you.

Answer

- Sociability levels did not change for those in the No Therapy group: $F(1,9) = .47$, $MSE = 0.42$, $p = .51$.
- There was a significant change in sociability for those in the CBT group, $F(1,9) = 17.05$, $MSE = 0.42$, $p = .003$, and in the sunshine group, $F(1,9) = 27.56$, $MSE = 0.36$, $p = .001$. Looking at the means in the *Descriptive Statistics* table, we can see that sociability went up for the CBT group and decreased in the sunshine group.

7 Versions

This document is available in [standard PDF](#), [simplified layout PDF](#), [standard dark theme PDF](#), [PDF with Open Dyslexic font](#), [large text PDF](#) and [spoken format](#). This document contains hyperlinks to [sections within it](#), [external webpages](#), and email addresses like ian.hocking@canterbury.ac.uk.

7.1 Licence

This work is licensed under a [Creative Commons](#) Attribution-NonCommercial-NoDerivatives 4.0 International License. The dark theme is based on [Solarised](#) by Ethan Schoonover.

8 References

- Field, A. (2009). *Discovering statistics using SPSS*. London: Sage Publications.
- Howell, D. C. (1997). *Statistical Methods for Psychology (Fourth Edition)*. California, USA: Wadsworth.